package org.papervision3d.objects.primitives {
	import org.papervision3d.core.geom.TriangleMesh3D;
	import org.papervision3d.core.geom.renderables.Triangle3D;
	import org.papervision3d.core.geom.renderables.Vertex3D;
	import org.papervision3d.core.proto.MaterialObject3D;	

	/**
	* <span lang="ja">PaperPlane は 3D オブジェクトの紙飛行機を作成するクラスです。
	* 紙飛行機はオブジェクトの方向を確認するテストの際に役立ちます。</span>
	* 
	* <span lang="en">The PaperPlane class lets you create a paper plane object.</span>
	* <p/>
	* <span lang="en">Paper planes are useful for testing, when you want to know the direction an object is facing.</span>
	*/
	public class PaperPlane extends TriangleMesh3D
	{
		/**
		* <span lang="ja">セグメントのデフォルト値です。</span>
		* <span lang="en">Default value of segments.</span>
		*/
		static public var DEFAULT_SCALE :Number = 1;
	
	
		// ___________________________________________________________________________________________________
		//                                                                                               N E W
		// NN  NN EEEEEE WW    WW
		// NNN NN EE     WW WW WW
		// NNNNNN EEEE   WWWWWWWW
		// NN NNN EE     WWW  WWW
		// NN  NN EEEEEE WW    WW
	
		/**
		* <span lang="ja">新しい PaperPlane インスタンスを作成します。</span>
		* <span lang="en">Creates a new PaperPlane object.</span>
		* @param	material	
		* <span lang="ja">テクスチャーとして適用したいMaterialObject3Dを指定します。</span>
		* <span lang="en">A Material3D object that contains the material properties of the object.</span>
		* @param	scale
		* <span lang="ja">スケール値です。</span>
		* <span lang="en">Scaling factor</span>
		* 
		* @example 次の例では PaperPlane を作成し、3Dシーンに PaperPlane を表示します。
		* <div class="listing"><pre>
		* import org.papervision3d.objects.primitives.PaperPlane;
		* import org.papervision3d.materials.WireframeMaterial;
		* 
		* // マテリアルを作成
		* var material:WireframeMaterial = new WireframeMaterial(0x0000FF);
		* 
		* // プリミティブオブジェクトを作成
		* var obj:PaperPlane = new PaperPlane(material, 2);
		* scene.addChild(obj);
		* </pre></div>
		* ※scene は BasicView クラスのプロパティー
		*/
		public function PaperPlane( material :MaterialObject3D=null, scale :Number=0 )
		{
			super( material, new Array(), new Array(), null );
	
			scale = scale || DEFAULT_SCALE;
	
			buildPaperPlane( scale );
		}
	
	
		private function buildPaperPlane( scale :Number ):void
		{
			var a :Number = 100 * scale;
			var b :Number = a/2;
			var c :Number = b/3;
	
			var v:Array =
			[
				new Vertex3D(  0,  0,  a ),
				new Vertex3D( -b,  c, -a ),
				new Vertex3D( -c,  c, -a ),
				new Vertex3D(  0, -c, -a ),
				new Vertex3D(  c,  c, -a ),
				new Vertex3D(  b,  c, -a )
			];
	
			this.geometry.vertices = v;
	
			this.geometry.faces =
			[
				new Triangle3D( this, [v[0], v[1], v[2]] ),
				new Triangle3D( this, [v[0], v[2], v[3]] ),
				new Triangle3D( this, [v[0], v[3], v[4]] ),
				new Triangle3D( this, [v[0], v[4], v[5]] )
			];
			
			this.projectTexture( "x", "z" );
	
			this.geometry.ready = true;
		}
	}
}
